Local Maxima, Minima, and Saddle points

Worksheet by Karin Burrell, Sang Lee

In this maple worksheet, we will be working with functions of several variables. In particular, we will learn how to find the local extreme values and saddle points of functions of two variables.

```
> restart;  with(plots) :
```

Introduction

Maple's **diff** command can be used to find the partial derivatives of a function of several variables. Type the following commands and observe each output carefully.

```
> f := (x, y) → x · y^2 − x^2 · y + 2 · x · y + 5;
  fx := diff( f(x, y), x);
  fy :=  diff( f(x, y), y);
```

$$f := (x, y) \to x\,y^2 - x^2\,y + 2\,y\,x + 5$$

$$fx := y^2 - 2\,y\,x + 2\,y$$

$$fy := 2\,y\,x - x^2 + 2\,x \qquad (1.1)$$

Do you agree with partial derivatives given by Maple?

If you want to compute higher order partial derivatives, do the following.

```
> fxx := diff( f(x, y), x$ 2);
  fyy := diff( f(x,  y), y$ 2);
```

$$fxx := -2\,y$$

$$fyy := 2\,x \qquad (1.2)$$

And for mixed partial derivatives;

```
> fxy := diff( f(x, y), x, y);
  fyx := diff( f(x, y), y, x);
```

$$fxy := 2\,y - 2\,x + 2$$

$$fyx := 2\,y - 2\,x + 2 \qquad (1.3)$$

To evaluate partial derivatives at a point, say at (1, 2), do the following commands.

```
> `fx(1, 2)` = subs( x = 1, y = 2, fx);
  `fy(1, 2)` = subs( x = 1, y = 2, fy);
  `fxx(1, 2)` = subs( x = 1, y = 2, fxx);
  `fyy(1, 2)` = subs( x = 1, y = 2, fyy);
  `fxy(1, 2)` = subs( x = 1, y = 2, fxy);
  `fyx(1, 2)` = subs( x = 1, y = 2, fyx);
```

$$fx(1,2) = 4$$
$$fy(1,2) = 5$$
$$fxx(1,2) = -4$$
$$fyy(1,2) = 2$$
$$fxy(1,2) = 4$$
$$fyx(1,2) = 4 \quad (1.4)$$

>

Graphical Example

Well, so much for the background. Let's look at an example.

Find all local extrema and saddle points of the function $f(x, y) = x^4 + y^4 - 4xy + 1$.

First, we enter the function into Maple and then look at its graph.

> $f := (x, y) \rightarrow x^4 + y^4 - 4 \cdot x \cdot y + 1;$
$plot3d(\ f(x, y), x = -2 ..2, y = -2 ..2, view = -5 ..8, axes = normal, style = patch);$

$$f := (x, y) \rightarrow x^4 + y^4 - 4\,y\,x + 1$$



If a function *f* is differentiable at local extrema, then the tangent plane must be horizontal. Click on the graph and rotate it until you see its important features.

Next, look at a contour map of the function.

> $f := x \rightarrow x^4 + y^4 - 4 \cdot x \cdot y + 1;$
> $contourplot(\; f(x, y), x = -2\,..2, y = -2\,..2, contours = [-2, -1, 0, 1, 2, 3, 4, 5], grid = [50, 50]);$

$$f := x \rightarrow x^4 + y^4 - 4\,x\,y + 1$$



In a contour map, a local extremum is located at a point where contour curves are concentric. On the other hand, saddle points are located where contour curves cross with the property that contours rise and fall alternately. In the above example, contour curves near (1, 1) and (-1, -1) are oval in shape and that as we move away from these points in any direction the values of function are increasing. So we predict the function f has local minimum at these points. However, as we move away from the origin (0, 0), the values of the function decrease in some directions and increase in other directions. Hence, we predict that the function has a saddle point at the origin.

Exercise

Graph the function and plot the contour curves of the function $f(x, y) = y^4 - x^3 - 2y^2 + 3x$ and then use them to predict the location of local extrema and the saddle points of f.

>

Numerical Example

> *restart*; *with*(*Spread*) :

We will now use a numerical approach to estimate the local extrema of our function $f(x, y) = x^4 + y^4 - 4xy + 1$. From our graph above, we decide to look at a sample of function values on the intervals x = -2..2 and y = -2..2.

> $f := (x, y) \rightarrow x^4 + y^4 - 4 \cdot x \cdot y + 1;$
> $xleft := -2; \quad xright := 2;$
> $yleft := -2; \quad yright := 2;$
> $increment := .5;$

$$f := (x, y) \rightarrow x^4 + y^4 - 4\,x\,y + 1$$
$$xleft := -2$$
$$xright := 2$$
$$yleft := -2$$
$$yright := 2$$
$$increment := 0.5 \qquad (3.1)$$

> $xdim := \operatorname{ceil}\left(\left(\frac{xright - xleft}{increment}\right) + 2\right);$
> $ydim := \operatorname{ceil}\left(\left(\frac{yright - yleft}{increment}\right) + 2\right);$

$$xdim := 10$$
$$ydim := 10 \qquad (3.2)$$

> $points := Matrix(1..ydim, 1..xdim):$
> $points[1, 1] := y \setminus x:$

for i **from** 2 **to** $xdim$ **do**
$points[1, i] := evalf\left(\left(\frac{xright - xleft}{xdim - 2}\right) \cdot (i - 2) + xleft\right):$
od:

for i **from** 2 **to** $ydim$ **do**
$points[i, 1] := evalf\left(\left(\frac{yright - yleft}{ydim - 2}\right) \cdot (i - 2) + yleft\right):$
od:

for k **from** 2 **to** $xdim$ **do**
for j **from** 2 **to** $ydim$ **do**
$points[j, k] := evalf(f(points[1, k], points[j, 1]));$
od:**od**:

$CreateSpreadsheet(data);$
$SetMatrix(data, points);$

	A	B	C	D	E	F
1	$y \setminus x$	−2.	−1.5000000(	−1.	−.50000000(	0.
2	−2.	17.	10.06250000	10.	13.06250000	17.
3	−1.5000000(	10.06250000	2.125000000	1.062500000	3.125000000	6.0625000
4	−1.	10.	1.062500000	−1.	0.062500000	2.
5	−.50000000(	13.06250000	3.125000000	0.062500000	0.12500000\	1.0625000
6	0.	17.	6.062500000	2.	1.062500000	1.

$$data \quad (3.3)$$

>

Now click on the spreadsheet that has been created. If you move your cursor to the lower right corner, you can expand the table to take a look at all the function values that have been calculated.

In the spreadsheet, the x values we used in our function are displayed along the first row. The y values are shown down the first column.

Looking at the table, estimate the points where the local extrema appear to be located.

Symbolic Example

To confirm our prediction on the local extrema and saddle points, we use the **second partial test**. Since the domain of $f(x, y)$ is the entire xy-plane, the local extrema and saddle points, if they exist, can occur only at the critical points. To find all the critical points, we set each first partial derivative equal to zero and solve for variables x and y.

```
> f := (x, y) -> x^4 + y^4 - 4·x·y + 1;
  fx := diff( f(x, y), x);
  fy := diff( f(x, y), y);
  `Critical points` = solve( { fx = 0, fy = 0}, {x, y} );
```

$$f := (x, y) \rightarrow x^4 + y^4 - 4\,x\,y + 1$$

$$fx := 4\,x^3 - 4\,y$$

$$fy := 4\,y^3 - 4\,x$$

$$Critical\ points = \big(\{y = 0, x = 0\}, \{x = -\mathrm{RootOf}(\_Z^2 + 1, label = \_L4), y = \mathrm{RootOf}(\_Z^2 + 1, label = \_L4)\}, \{x = 1, y = 1\}, \{y = -1, x = -1\}, \{y = \mathrm{RootOf}(-\mathrm{RootOf}(\_Z^2 + 1, label = \_L3) + \_Z^2, label = \_L5), x = \mathrm{RootOf}(-\mathrm{RootOf}(\_Z^2 + 1, label = \_L3) + \_Z^2, label = \_L5)\ \mathrm{RootOf}(\_Z^2 + 1, label = \_L3)\}\big) \quad (4.1)$$

In this example, Maple was able to solve the system of equations and gives coordinates of three critical points; (0, 0), (1, 1), and (-1, -1).
Next, we want to apply the **second partial test**. Thus we determine the discriminant ***D*** at each critical point.

$$\boldsymbol{D} = f_{xx}\, f_{yy} - \left[\, f_{xy} \,\right]^2$$

> $Disc := diff(\, f(x,\, y),\, x\$\, 2) \cdot diff(\, f(x,y),\, y\$\, 2) - diff(\, f(x,y),\, x,\, y)^2;$

$$Disc := 144\, x^2\, y^2 - 16 \qquad (4.2)$$

We now substitute the coordinates of each of the critical points into "Disc".

> $subs(\, x=0,\ y=0,\ Disc);$
> $subs(\, x=1, y=1, Disc);$
> $subs(\, x=-1, y=-1, Disc);$

$$-16$$
$$128$$
$$128 \qquad (4.3)$$

Since the discriminant Disc is negative at $(0, 0)$, the function $f(x, y)$ has a saddle point at $(0, 0)$.

> $f(0, 0);$

$$1 \qquad (4.4)$$

Therefore, the point $(0, 0, 1)$ is a **saddle point** on the surface.

We see that the values of Disc at $(1, 1)$ and $(-1, -1)$ are positive. So, we must now check the sign of the second partial derivative (w.r.t. x) at these two points.

> $fxx := diff(\, f(x,\, y),\, x\$\, 2);$
> $subs(\, x=1, y=1, fxx);$
> $subs(\, x=-1, y=-1, fxx);$

$$fxx := 12\, x^2$$
$$12$$
$$12 \qquad (4.5)$$

As we can see $f_{xx} > 0$ at both $(1, 1)$ and $(-1, -1)$ and hence we conclude that $f(x, y)$ has a **local minimum** at $(1, 1)$ and $(-1, -1)$.
The **minimum value** at these points are

> $subs(\, x=1, y=1, f(x, y));$
> $subs(\, x=-1, y=-1, f(x, y));$

$$-1$$
$$-1 \qquad (4.6)$$

Exercises

(1) Use a graph and contour curves to estimate the local maximum and minimum values and saddle points of the function $f(x, y) = y^3 + 3x^2y - 3x^2 - 3y^2 + 2$.
Then use the second partial test to find these values precisely.

>

(2) Find the volume of the largest rectangular box with edges parallel to the axes that can be inscribed in the ellipsoid $9x^2 + 36y^2 + 4z^2 = 36$.

>